

Functional programming languages

Part IV: monadic transformations, monadic programming

Xavier Leroy

INRIA Rocquencourt

MPRI 2-4-2, 2007

Monads in programming language theory

Monads are a technical device with several uses in programming:

- To structure denotational semantics and make them easy to extend with new language features. (E. Moggi, 1989.)
Not treated in this lecture.
- To factor out commonalities between many program transformations and between their proofs of correctness.
- As a powerful programming techniques in pure functional languages. (P. Wadler, 1992; the Haskell community).

Outline

- 1 Introduction to monads
- 2 The monadic translation
 - Definition
 - Correctness
 - Application to some monads
- 3 Monadic programming
 - More examples of monads
 - Monad transformers

Commonalities between program transformations

Consider the conversions to exception-returning style, state-passing style, and continuation-passing style. For constants, variables and λ -abstractions, we have:

$$\begin{array}{lll} \llbracket N \rrbracket = V(N) & \llbracket N \rrbracket = \lambda s.(N, s) & \llbracket N \rrbracket = \lambda k.k N \\ \llbracket x \rrbracket = V(x) & \llbracket x \rrbracket = \lambda s.(x, s) & \llbracket x \rrbracket = \lambda k.k x \\ \llbracket \lambda x.a \rrbracket = V(\lambda x.\llbracket a \rrbracket) & \llbracket \lambda x.a \rrbracket = \lambda s.(\lambda x.\llbracket a \rrbracket, s) & \llbracket \lambda x.a \rrbracket = \lambda k.k (\lambda x.\llbracket a \rrbracket) \end{array}$$

in all three cases, we **return** (put in some appropriate wrapper) the values N or x or $\lambda x.\llbracket a \rrbracket$.

Commonalities between program transformations

For let bindings, we have:

$$\begin{aligned} \llbracket \text{let } x = a \text{ in } b \rrbracket &= \text{match } \llbracket a \rrbracket \text{ with } E(x) \rightarrow E(x) \mid V(x) \rightarrow \llbracket b \rrbracket \\ \llbracket \text{let } x = a \text{ in } b \rrbracket &= \lambda s. \text{match } \llbracket a \rrbracket s \text{ with } (x, s') \rightarrow \llbracket b \rrbracket s' \\ \llbracket \text{let } x = a \text{ in } b \rrbracket &= \lambda k. \llbracket a \rrbracket (\lambda x. \llbracket b \rrbracket k) \end{aligned}$$

In all three cases, we extract (one way or another) the value contained in the computation $\llbracket a \rrbracket$, **bind** it to the variable x , and proceed with the computation $\llbracket b \rrbracket$.

Commonalities between program transformations

Concerning function applications:

$$\begin{aligned} \llbracket a \ b \rrbracket &= \text{match } \llbracket a \rrbracket \text{ with} \\ &\quad \mid E(e_a) \rightarrow E(e_a) \\ &\quad \mid V(v_a) \rightarrow \\ &\quad \quad \text{match } \llbracket b \rrbracket \text{ with } E(e_b) \rightarrow E(e_b) \mid V(v_b) \rightarrow v_a \ v_b \\ \llbracket a \ b \rrbracket &= \lambda s. \text{match } \llbracket a \rrbracket s \text{ with } (v_a, s') \rightarrow \\ &\quad \text{match } \llbracket b \rrbracket s' \text{ with } (v_b, s'') \rightarrow v_a \ v_b \ s'' \\ \llbracket a \ b \rrbracket &= \lambda k. \llbracket a \rrbracket (\lambda v_a. \llbracket b \rrbracket (\lambda v_b. v_a \ v_b \ k)) \end{aligned}$$

We bind $\llbracket a \rrbracket$ to a variable v_a , then bind $\llbracket b \rrbracket$ to a variable v_b , then perform the application $v_a \ v_b$.

Interface of a monad

A monad is defined by a parameterized type $\alpha \text{ mon}$ and operations `ret`, `bind` and `run`, with types:

$$\begin{aligned} \text{ret} & : \forall \alpha. \alpha \rightarrow \alpha \text{ mon} \\ \text{bind} & : \forall \alpha, \beta. \alpha \text{ mon} \rightarrow (\alpha \rightarrow \beta \text{ mon}) \rightarrow \beta \text{ mon} \\ \text{run} & : \forall \alpha. \alpha \text{ mon} \rightarrow \alpha \end{aligned}$$

The type $\tau \text{ mon}$ is the type of **computations** that eventually produce a value of type τ .

`ret a` encapsulates a pure expression $a : \tau$ as a trivial computation (of type $\tau \text{ mon}$) that immediately produces the value of a .

`bind a (\lambda x. b)` performs the computation $a : \tau \text{ mon}$, binds its value to $x : \tau$, then performs the computation $b : \tau' \text{ mon}$.

`run a` is the execution of a whole monadic program a , extracting its return value.

Monadic laws

The `ret` and `bind` operations of the monad are supposed to satisfy the following algebraic laws:

$$\begin{aligned} \text{bind} (\text{ret } a) f & \approx f a \\ \text{bind } a (\lambda x. \text{ret } x) & \approx a \\ \text{bind} (\text{bind } a (\lambda x. b)) (\lambda y. c) & \approx \text{bind } a (\lambda x. \text{bind } b (\lambda y. c)) \end{aligned}$$

The relation \approx needs to be made more precise, but intuitively means “behaves identically”.

Example: the Exception monad

(also called the Error monad)

```
type  $\alpha$  mon = V of  $\alpha$  | E of exn
```

```
ret a = V(a)
```

```
bind m f = match m with E(x) -> E(x) | V(x) -> f x
```

```
run m = match m with V(x) -> x
```

bind encapsulates the propagation of exceptions in compound expressions such as *a b* or let bindings.

Additional operations in this monad:

```
raise x = E(x)
```

```
trywith m f = match m with E(x) -> f x | V(x) -> V(x)
```

Example: the State monad

```
type  $\alpha$  mon = state  $\rightarrow$   $\alpha \times$  state
```

```
ret a =  $\lambda$ s. (a, s)
```

```
bind m f =  $\lambda$ s. match m s with (x, s') -> f x s'
```

```
run m = match m empty_store with (x, s) -> x
```

bind encapsulates the threading of the state in compound expressions.

Additional operations in this monad:

```
ref x =  $\lambda$ s. store_alloc x s
```

```
deref r =  $\lambda$ s. (store_read r s, s)
```

```
assign r x =  $\lambda$ s. store_write r x s
```

Example: the Continuation monad

```
type  $\alpha$  mon = ( $\alpha \rightarrow$  answer)  $\rightarrow$  answer
```

```
ret a =  $\lambda k.$  k a
```

```
bind m f =  $\lambda k.$  m ( $\lambda v.$  f v k)
```

```
run m = m ( $\lambda x.$  x)
```

Additional operations in this monad:

```
callcc f =  $\lambda k.$  f k k
```

```
throw x y =  $\lambda k.$  x y
```

Outline

- 1 Introduction to monads
- 2 **The monadic translation**
 - Definition
 - Correctness
 - Application to some monads
- 3 Monadic programming
 - More examples of monads
 - Monad transformers

The monadic translation

Core constructs

$$\begin{aligned}
 \llbracket N \rrbracket &= \text{ret } N \\
 \llbracket x \rrbracket &= \text{ret } x \\
 \llbracket \lambda x. a \rrbracket &= \text{ret } (\lambda x. \llbracket a \rrbracket) \\
 \llbracket \text{let } x = a \text{ in } b \rrbracket &= \text{bind } \llbracket a \rrbracket (\lambda x. \llbracket b \rrbracket) \\
 \llbracket a \ b \rrbracket &= \text{bind } \llbracket a \rrbracket (\lambda v_a. \text{bind } \llbracket b \rrbracket (\lambda v_b. v_a \ v_b))
 \end{aligned}$$

These translation rules are shared between all monads.

Effect on types: if $a : \tau$ then $\llbracket a \rrbracket : \llbracket \tau \rrbracket \text{ mon}$

where $\llbracket \tau_1 \rightarrow \tau_2 \rrbracket = \tau_1 \rightarrow \llbracket \tau_2 \rrbracket \text{ mon}$ and $\llbracket \tau \rrbracket = \tau$ for base types τ .

The monadic translation

Extensions

$$\begin{aligned}
 \llbracket \mu f. \lambda x. a \rrbracket &= \text{ret } (\mu f. \lambda x. \llbracket a \rrbracket) \\
 \llbracket a \ \text{op} \ b \rrbracket &= \text{bind } \llbracket a \rrbracket (\lambda v_a. \text{bind } \llbracket b \rrbracket (\lambda v_b. \text{ret } (v_a \ \text{op} \ v_b))) \\
 \llbracket C(a_1, \dots, a_n) \rrbracket &= \text{bind } \llbracket a_1 \rrbracket (\lambda v_1. \dots \\
 &\quad \text{bind } \llbracket a_n \rrbracket (\lambda v_n. \text{ret}(C(v_1, \dots, v_n))))
 \end{aligned}$$

$$\llbracket \text{match } a \text{ with } \dots p_i \dots \rrbracket = \text{bind } \llbracket a \rrbracket (\lambda v_a. \text{match } v_a \text{ with } \dots \llbracket p_i \rrbracket \dots)$$

$$\llbracket C(x_1, \dots, x_n) \rightarrow a \rrbracket = C(x_1, \dots, x_n) \rightarrow \llbracket a \rrbracket$$

Example of monadic translation

```

[[1 + f x]] =
  bind (ret 1) (\v1.
    bind (bind (ret f) (\v2.
      bind (ret x) (\v3. v2 v3))) (\v4.
    ret (v1 + v4)))

```

After administrative reductions using the first monadic law:

```

[[1 + f x]] =
  bind (f x) (\v. ret (1 + v))

```

Example of monadic translation

```

[[μfact. λn. if n = 0 then 1 else n * fact(n-1)]] =
  ret (μfact. λn.
    if n = 0
    then ret 1
    else bind (fact(n-1)) (\v. ret (n * v))

```


The monadic translation

Monad-specific constructs and operations

Most additional constructs for exceptions, state and continuations can be treated as regular function applications of the corresponding additional operations of the monad. For instance, in the case of `raise a`:

$$\begin{aligned} \llbracket \text{raise } a \rrbracket &= \text{bind } (\text{ret } \text{raise}) (\lambda v_r. \text{bind } \llbracket a \rrbracket (\lambda v_a. v_r v_a)) \\ &\xrightarrow{\text{adm}} \text{bind } \llbracket a \rrbracket (\lambda v_a. \text{raise } v_a) \end{aligned}$$

The `bind` takes care of propagating exceptions raised in `a`.

The only case where we need a special translation rule is the the `try...with` construct:

$$\llbracket \text{try } a \text{ with } x \rightarrow b \rrbracket = \text{trywith } \llbracket a \rrbracket (\lambda x. \llbracket b \rrbracket)$$

Syntactic properties of the monadic translation

Define the monadic translation of a value $\llbracket v \rrbracket_v$ as follows:

$$\llbracket N \rrbracket_v = N \quad \llbracket \lambda x. a \rrbracket_v = \lambda x. \llbracket a \rrbracket$$

Lemma 1 (Translation of values)

$\llbracket v \rrbracket = \text{ret } \llbracket v \rrbracket_v$ for all values v . Moreover, $\llbracket v \rrbracket_v$ is a value.

Lemma 2 (Monadic substitution)

$\llbracket a[x \leftarrow v] \rrbracket = \llbracket a \rrbracket[x \leftarrow \llbracket v \rrbracket_v]$ for all values v ,

Reasoning about reductions of the translations

If a reduces, is it the case that the translation $\llbracket a \rrbracket$ reduces? This depends on the monad:

- For the exception monad, this is true.
- For the state and continuation monads, $\llbracket a \rrbracket$ is a λ -abstraction which cannot reduce.

To reason about the evaluation of $\llbracket a \rrbracket$, we need in general to put this term in an appropriate **context**, for instance

- For the state monad: $\llbracket a \rrbracket s$ where s is a store value.
- For the continuation monad: $\llbracket a \rrbracket k$ where k is a continuation $\lambda x \dots$

Contextual equivalence

To overcome this problem, we assume that the monad defines an equivalence relation $a \approx a'$ between terms, which is reflexive, symmetric and transitive, and satisfies the following properties:

- 1 $(\lambda x. a) v \approx a[x \leftarrow v]$
- 2 $\text{bind} (\text{ret } v) (\lambda x. b) \approx b[x \leftarrow v]$
- 3 $\text{bind } a (\lambda x. b) \approx \text{bind } a' (\lambda x. b)$ if $a \approx a'$
- 4 If $a \approx \text{ret } v$, then $\text{run } a \xrightarrow{*} v$.

Correctness of the monadic translation

Theorem 3

If $a \Rightarrow v$, then $\llbracket a \rrbracket \approx \text{ret } \llbracket v \rrbracket_v$.

The proof is by induction on a derivation of $a \Rightarrow v$ and case analysis on the last evaluation rule.

The cases $a = N$, $a = x$ and $a = \lambda x.b$ are obvious: we have $a = v$, therefore $\llbracket a \rrbracket = \text{ret } \llbracket v \rrbracket_v$.

Correctness of the monadic translation

For the `let` case:

$$\frac{b \Rightarrow v' \quad c[x \leftarrow v'] \Rightarrow v}{\text{let } x = b \text{ in } c \Rightarrow v}$$

The following equivalences hold:

$$\begin{aligned} \llbracket a \rrbracket &= \text{bind } \llbracket b \rrbracket (\lambda x. \llbracket c \rrbracket) \\ (\text{ind.hyp} + \text{prop.3}) &\approx \text{bind } (\text{ret } \llbracket v' \rrbracket_v) (\lambda x. \llbracket c \rrbracket) \\ (\text{prop.2}) &\approx \llbracket c \rrbracket [x \leftarrow \llbracket v' \rrbracket_v] = \llbracket c[x \leftarrow v'] \rrbracket \\ (\text{ind.hyp.}) &\approx \text{ret } \llbracket v \rrbracket_v \end{aligned}$$

Correctness of the monadic translation

For the application case:

$$\frac{b \Rightarrow \lambda x.d \quad c \Rightarrow v' \quad d[x \leftarrow v'] \Rightarrow v}{b \ c \Rightarrow v}$$

The following equivalences hold:

$$\begin{aligned} \llbracket a \rrbracket &= \text{bind } \llbracket b \rrbracket (\lambda y. \text{bind } \llbracket c \rrbracket (\lambda z. y \ z)) \\ (\text{ind.hyp} + \text{prop.3}) &\approx \text{bind } (\text{ret } (\lambda x. \llbracket d \rrbracket)) (\lambda y. \text{bind } \llbracket c \rrbracket (\lambda z. y \ z)) \\ (\text{prop.2}) &\approx \text{bind } \llbracket c \rrbracket (\lambda z. (\lambda x. \llbracket d \rrbracket) \ z) \\ (\text{ind.hyp} + \text{prop.3}) &\approx \text{bind } (\text{ret } \llbracket v' \rrbracket_v (\lambda z. (\lambda x. \llbracket d \rrbracket) \ z)) \\ (\text{prop.2}) &\approx (\lambda x. \llbracket d \rrbracket) \llbracket v' \rrbracket_v \\ (\text{prop.1}) &\approx \llbracket d \rrbracket[x \leftarrow \llbracket v' \rrbracket_v] = \llbracket d[x \leftarrow v] \rrbracket \\ (\text{ind.hyp.}) &\approx \text{ret } \llbracket v \rrbracket_v \end{aligned}$$

Correctness of the monadic translation

Theorem 4

If $a \Rightarrow N$, then $\text{run } \llbracket a \rrbracket \xrightarrow{*} N$.

Proof.

Follows from theorem 3 and property 4 of \approx . □

Note that we proved this theorem only for pure terms a that do not use monad-specific constructs. These constructs add more cases, but often the proof cases for application, etc, are unchanged. (Exercise.)

Application to the Exception monad

Define $a_1 \approx a_2$ as $\exists a, a_1 \xrightarrow{*} a \xleftarrow{*} a_2$.

Some interesting properties of this relation:

- If $a \rightarrow a'$ then $a \approx a'$.
- If $a \approx a'$ and $a \xrightarrow{*} v$, then $a' \xrightarrow{*} v$.
- It is transitive, for if $a_1 \xrightarrow{*} a \xleftarrow{*} a_2 \xrightarrow{*} a' \xleftarrow{*} a_3$, determinism of the \rightarrow reduction implies that either $a \xrightarrow{*} a'$ or $a' \xrightarrow{*} a$. In the former case, $a_1 \xrightarrow{*} a' \xleftarrow{*} a_3$, and in the latter case, $a_1 \xrightarrow{*} a \xleftarrow{*} a_3$.
- It is compatible with reduction contexts: $E[a_1] \approx E[a_2]$ if $a_1 \approx a_2$ and E is a reduction context.

We now check that \approx satisfies the hypothesis of theorem 3.

Application to the Exception monad

- 1 $(\lambda x.a) v \approx a[x \leftarrow v]$
Trivial since $(\lambda x.a) v \rightarrow a[x \leftarrow v]$.
- 2 $\text{bind}(\text{ret } v) (\lambda x.b) \approx b[x \leftarrow v]$. We have

$$\begin{aligned} & \text{bind}(\text{ret } v) (\lambda x.b) \\ & \rightarrow \text{bind}(V(v)) (\lambda x.b) \\ & \xrightarrow{*} \text{match } V(v) \text{ with } E(y) \rightarrow y \mid V(z) \rightarrow (\lambda x.b) z \\ & \rightarrow (\lambda x.b) v \rightarrow b[x \leftarrow v] \end{aligned}$$

- 3 $\text{bind } a_1 (\lambda x.b) \approx \text{bind } a_2 (\lambda x.b)$ if $a_1 \approx a_2$.
Trivial since $\text{bind } [] (\lambda x.b)$ is an evaluation context.
- 4 If $a \approx \text{ret } v$, then $\text{run } a \xrightarrow{*} v$.
Since $\text{ret } v \xrightarrow{*} V(v)$, we have $a \xrightarrow{*} V(v)$ and the result follows.

Application to the Continuation monad

Define $a_1 \approx a_2$ as $\forall k \in \text{Values}, \exists a, a_1 k \xrightarrow{*} a \xleftarrow{*} a_2 k$.

- ① $(\lambda x.a) v \approx a[x \leftarrow v]$
Trivial since $(\lambda x.a) v k \rightarrow a[x \leftarrow v] k$.
- ② $\text{bind} (\text{ret } v) (\lambda x.b) \approx b[x \leftarrow v]$. We have

$$\begin{aligned}
 \text{bind} (\text{ret } v) (\lambda x.b) k &\rightarrow \text{bind} (\lambda k'. k' v) (\lambda x.b) \\
 &\xrightarrow{*} (\lambda k'. k' v) (\lambda y. (\lambda x.b) y k) \\
 &\rightarrow (\lambda y. (\lambda x.b) y k) v \\
 &\rightarrow (\lambda x.b) v k \\
 &\rightarrow b[x \leftarrow v] k
 \end{aligned}$$

Application to the Continuation monad

- ① $\text{bind } a_1 (\lambda x.b) \approx \text{bind } a_2 (\lambda x.b)$ if $a_1 \approx a_2$
We have $\text{bind } a_i (\lambda x.b) k \xrightarrow{*} a_i (\lambda v. (\lambda x.b) v k)$ for $i = 1, 2$.
Using the hypothesis $a_1 \approx a_2$ with the continuation $(\lambda v. (\lambda x.b) v k)$,
we obtain a term a such that $a_i (\lambda v. (\lambda x.b) v k) \xrightarrow{*} a$ for $i = 1, 2$.
Therefore, $\text{bind } a_i (\lambda x.b) k \xrightarrow{*} a$ for $i = 1, 2$, and the result follows.
- ② If $a \approx \text{ret } v$, then $\text{run } a \xrightarrow{*} v$.
The result follows from $\text{ret } v (\lambda x.x) \xrightarrow{*} v$.

Application to the State monad

Define $a_1 \approx a_2$ as $\forall s \in \text{Values}, \exists a, a_1 s \xrightarrow{*} a \xleftarrow{*} a_2 s$.

The proofs of hypotheses 1–4 are similar to those for exceptions.

Outline

- 1 Introduction to monads
- 2 The monadic translation
 - Definition
 - Correctness
 - Application to some monads
- 3 Monadic programming
 - More examples of monads
 - Monad transformers

Monads as a general programming technique

Monads provide a systematic way to **structure** programs into two well-separated parts:

- the algorithms proper, and
- the “plumbing” of computations needed by these algorithms (state passing, exception handling, non-deterministic choice, etc).

In addition, monads can also be used to **modularize** code and offer new possibilities for reuse:

- Code in monadic form can be parameterized over a monad and reused with several monads.
- Monads themselves can be built in an incremental manner.

The Logging monad (a.k.a. the Writer monad)

Enables computations to log messages. A special case of the State monad, guaranteeing that the log grows monotonically.

```

module Log = struct
  type log = string list
  type  $\alpha$  mon = log  $\rightarrow$   $\alpha \times$  log

  let ret a = fun l -> (a, l)
  let bind m f = fun l -> match m l with (x, l') -> f x l'
  let run m = match m [] with (x, l) -> (x, List.rev l)

  let log msg = fun l -> ((), msg :: l)
end

```


Example of use

Before monadic translation:

```
let abs n =
  if n >= 0
  then (log "positive"; n)
  else (log "negative"; -n)
```

After monadic translation:

```
let abs n =
  if n >= 0
  then Log.bind (Log.log "positive") (fun _ -> Log.ret n)
  else Log.bind (Log.log "negative") (fun _ -> Log.ret (-n))
```

Non-determinism, a.k.a. the List monad

Provides computations with non-deterministic choice as well as failure. Underneath, computes the list of all possible results.

```
module Nondet = struct
  type  $\alpha$  mon =  $\alpha$  list

  let ret a = a :: []
  let rec bind m f =
    match m with [] -> [] | hd :: tl -> f hd @ bind tl f
  let run m = match m with hd :: tl -> hd
  let runall m = m

  let fail = []
  let either a b = a @ b
end
```

Example of use

All possible ways to insert an element x in a list l :

```
let rec insert x l =
  Nondet.either (Nondet.ret (x :: l))
    (match l with
     | [] -> Nondet.fail
     | hd :: tl ->
       Nondet.bind (insert x tl)
                   (fun l' -> Nondet.ret (hd :: l'))))
```

All permutations of a list l :

```
let rec permut l =
  match l with
  | [] -> Nondet.ret []
  | hd :: tl ->
    Nondet.bind (permut tl) (fun l' -> insert hd l')
```

Monads for randomized computations

Consider a source language with randomized constructs such as

<code>rand n</code>	return a uniformly-distributed integer in $[0, n[$
<code>choose p a b</code>	evaluate a with probability $p \in [0, 1]$ evaluate b with probability $1 - p$

In a monadic interpretation, these constructs have type

```
rand   : int → int mon
choose : ∀α. float → α mon → α mon → α mon
```

Examples of randomized computations

```

let roll_3d6 =
  M.bind (M.rand 6) (fun d1 ->
    M.bind (M.rand 6) (fun d2 ->
      M.bind (M.rand 6) (fun d3 ->
        M.return (1+d1 + 1+d2 + 1+d3))))))

let traffic_light =
  M.choose 0.05 (M.return Yellow)
    (M.choose 0.5 (M.return Red)
      (M.return Green))

```

First implementation: the Simulation monad

Uses a pseudo-random number generator to give values to random variables (Monte-Carlo simulation). This is a variant of the State monad.

```

module Random_Simulation = struct
  type  $\alpha$  mon = int  $\rightarrow$   $\alpha \times$  int

  let ret a = fun s -> (a, s)
  let bind m f = fun s -> match m s with (x, s) -> f x s

  let next_state s = s * 25173 + 1725
  let rand n = fun s -> ((abs s) mod n, next_state s)
  let choose p a b = fun s ->
    if float (abs s) <= p *. float max_int
    then a (next_state s) else b (next_state s)
end

```

Second implementation: the Distribution monad

With the same interface, this monad computes the distribution of the results: all possible result values along with their probabilities.

This is an extension of the List monad.

```

module Random_Distribution = struct
  type  $\alpha$  mon = ( $\alpha \times \text{float}$ ) list

  let ret a = [(a, 1.0)]
  let bind m f =
    [ (y, p1 *. p2) | (x, p1) <- m, (y, p2) <- f x ]

  let rand n = [ (0,  $\frac{1}{n}$ ); ...; (n-1,  $\frac{1}{n}$ ) ]
  let choose p a b =
    [ (x, p *. p1) | (x, p1) <- a ] @
    [ (x, (1.0 -. p) *. p2) | (x, p2) <- b ]
end

```

Third implementation: the Expectation monad

Still with the same interface, this monad computes the expectation of a result (of type α) w.r.t. a given measure (a function $\alpha \rightarrow \text{float}$).

This is an extension of the Continuation monad.

```

module Random_Expectation = struct
  type  $\alpha$  mon = ( $\alpha \rightarrow \text{float}$ )  $\rightarrow$  float

  let ret x = fun k  $\rightarrow$  k x
  let bind x f = fun k  $\rightarrow$  x (fun vx  $\rightarrow$  f vx k)

  let rand n = fun k  $\rightarrow$   $\frac{1}{n}$  *. k 0 +. ... +.  $\frac{1}{n}$  *. k (n-1)

  let choose p a b = fun k  $\rightarrow$  p *. a k +. (1.0 -. p) *. b k
end

```

Combining monads

What if we need both exceptions and state in an algorithm?

We can write (from scratch) a monad that supports both. Notice that there are several choices:

- type $\alpha \text{ mon} = \text{state} \rightarrow (\alpha \times \text{state}) \text{ outcome}$
i.e. the state is discarded when we raise an exception.
- type $\alpha \text{ mon} = \text{state} \rightarrow \alpha \text{ outcome} \times \text{state}$
i.e. the state is kept when we raise an exception.

In the second case, `trywith` can be defined in two ways:

$$\begin{aligned} \text{trywith } m \ f \ = \ \lambda s. \text{ match } m \ s \ \text{with} \\ & \quad | (V(v), s') \rightarrow (V(v), s') \\ & \quad | (E(e), s') \rightarrow f \ e \ \begin{pmatrix} s \\ s' \end{pmatrix} \end{aligned}$$

The s choice backtracks the assignments made by the computation m ; the s' choice preserves them.

Monad transformers

A more systematic way to build combined monads is to use **monad transformers**.

A monad transformer takes any monad M and returns a monad M' with additional capabilities, e.g. exceptions, state, continuation. It also provides a `lift` function that transforms M computations (of type $\alpha \ M.\text{mon}$) into M' computations (of type $\alpha \ M'.\text{mon}$)

In Caml, monad transformers are naturally presented as **functors**, i.e. functions from modules to modules. (Haskell uses type classes.)

Signature for monads

The Caml module signature for a monad is:

```
module type MONAD = sig
  type  $\alpha$  mon
  val ret:  $\alpha \rightarrow \alpha$  mon
  val bind:  $\alpha$  mon  $\rightarrow$  ( $\alpha \rightarrow \beta$  mon)  $\rightarrow \beta$  mon
  val run:  $\alpha$  mon  $\rightarrow \alpha$ 
end
```

The Identity monad

The Identity monad is a trivial instance of this signature:

```
module Identity = struct
  type  $\alpha$  mon =  $\alpha$ 
  let ret x = x
  let bind m f = f m
  let run m = m
end
```

Monad transformer for exceptions

```
module ExceptionTransf(M: MONAD) = struct
  type  $\alpha$  outcome = V of  $\alpha$  | E of exn
  type  $\alpha$  mon = ( $\alpha$  outcome) M.mon

  let ret x = M.ret (V x)
  let bind m f =
    M.bind m (function E e -> M.ret (E e) | V v -> f v)
  let lift x = M.bind x (fun v -> M.ret (V v))
  let run m = M.run (M.bind m (function V x -> M.ret x))

  let raise e = M.ret (E e)
  let trywith m f =
    M.bind m (function E e -> f e | V v -> M.ret (V v))
end
```

Monad transformer for state

```
module StateTransf(M: MONAD) = struct
  type  $\alpha$  mon = state -> ( $\alpha$  * state) M.mon

  let ret x = fun s -> M.ret (x, s)
  let bind m f =
    fun s -> M.bind (m s) (fun (x, s') -> f x s')
  let lift m = fun s -> M.bind m (fun x -> M.ret (x, s))
  let run m =
    M.run (M.bind (m empty_store) (fun (x, s') -> M.ret x))

  let ref x = fun s -> M.ret (store_alloc x s)
  let deref r = fun s -> M.ret (store_read r s, s)
  let assign r x = fun s -> M.ret (store_write r x s)
end
```

Monad transformer for continuations

```

module ContTransf(M: MONAD) = struct
  type  $\alpha$  mon = ( $\alpha$  -> answer M.mon) -> answer M.mon

  let ret x = fun k -> k x
  let bind m f = fun k -> m (fun v -> f v k)
  let lift m = fun k -> M.bind m k
  let run m = M.run (m (fun x -> M.ret x))

  let callcc f = fun k -> f k k
  let throw c x = fun k -> c x
end

```

Using monad transformers

```

module StateAndException = struct
  include ExceptionTransf(State)
  let ref x = lift (State.ref x)
  let deref r = lift (State.deref r)
  let assign r x = lift (State.assign r x)
end

```

This gives a type α mon = state \rightarrow α outcome \times state, i.e. state is preserved when raising exceptions.

The other combination, StateTransf(Exception) gives α mon = state \rightarrow ($\alpha \times$ state) outcome, i.e. state is discarded when an exception is raised.

The Concurrency monad transformer

Generalizing the Continuation monad transformer, we can define concurrency (interleaving of atomic computations) as follows:


```

module Concur(M: MONAD) = struct
  type answer =
    | Seq of answer M.mon
    | Par of answer * answer
    | Stop
  type  $\alpha$  mon = ( $\alpha$  -> answer) -> answer
  let ret x = fun k -> k x
  let bind x f = fun k -> x (fun v -> f v k)
  let atom m = fun k -> Seq(M.bind m (fun v -> M.ret (k v)))
  let stop = fun k -> Stop
  let par m1 m2 = fun k -> Par (m1 k, m2 k)

```

The Concurrency monad transformer

If $m : \alpha$ mon, applying m to the initial continuation $\lambda x, \text{Stop}$ builds a tree of computations such as:



All that remains is to execute the atomic actions m_1, \dots, m_6 in breadth-first order, simulating interleaved execution.

The Concurrency monad transformer

```

module Concur(M: MONAD) = struct
  ...
  let rec schedule acts =
    match acts with
    | [] -> M.ret ()
    | Seq m :: rem ->
      M.bind m (fun m' -> schedule (rem @ [m']))
    | Par(a1, a2) :: rem ->
      schedule (a1 :: a2 :: rem)
    | Stop :: rem ->
      schedule rem
  let run m = M.run (schedule [m (fun _ -> Stop)])
end

```

Example of use

```

module M = Concur(Log)

let rec loop n s =
  if n <= 0
  then M.ret ()
  else M.bind (M.atom (Log.log s)) (fun _ -> loop (n-1) s)

M.run (M.bind (M.atom (Log.log "start:")) (fun _ ->
  M.par (loop 6 "a") (loop 4 "b")))

```

This code will log “start:ababababaaaa”